

Calculating Solenoid Force Using Macro Elements

David B. Mohler
email: djdm@woh.rr.com

Abstract-This article shows how solenoid forces are calculated without finite element analysis or cad software using a strictly mathematical model description to solve the magnetics of a d.c. solenoid.

1. - Introduction

It has been observed by inquiries and discussions online that the interest in solenoid magnetics is alive and well but without a clear concept as how to deal with the non-linearities that are attendant in solving the solenoid model for force output. For the inquiring individual this paper now immediately jumps into the explanation where it may be found that it is both surprisingly simple yet challenging. Later, some observations and comments will be made on this subject.

2 - Key to Solving The Magnetics

Suppose a magnetic flux is passed (propagated) uniformly thru a ferrous tube as shown in Fig. 1. Let the tube dimensions, o.d., i.d., and length, be known as well as the type of steel. The presence of a flux within the steel causes an mmf drop (ampere-turns) across the length of the tube (one of several macro elements of the solenoid). This value of mmf is easily calculated.




Fig. 1 Flux Passing Thru a Ferrous Tube

The flux density for the tube is simply flux (webers) divided by the path area, $B = \phi / A$. From the steel's BH characteristics, H is found from a curve or lookup table or equation. (We will use an equation). Having found H (ampere turns/meter), then the mmf drop = H x length of the tube ($\mathfrak{S} = H\ell$). In a solenoid the flux conducts as a continuous loop following the path of the ferrous components; the loop is not circumferential but rather longitudinal thru the case tube (as shown), plunger, stator pole, main working air gap, across the clearance air gap, and radially thru the end pieces (from tube to poles). Thus the flux is virtually the same value thru each component and air gap feature in a series path. At the outset of a magnetic circuit calculation, the starting flux is set as a completely arbitrary value. By calculating the value of flux density (B) for each component and its corresponding H, the mmf drops are summed including those for air gap drops. Next, the value of flux is adjusted up or down, as appropriate, and the calculations reiterated until the summed mmf drops is equal to the coil's ampere turns (which is a constant throughout the calculation). This then defines the true solenoid flux relating to the starting conditions of coil ampere turns and plunger position as will be illustrated in the computer routine that follows.. The solenoid force will be derived from this level of flux in the working air gap and its mmf drop, \mathfrak{S}_g .

The relationship of H to B for the steel is fundamentally important. The BH curve is non-linear and each component (macro element), due to differences in geometry and flux density, will have a unique value of H. Equation (1) provides a polynomial equation for a typical grade of solenoid steel for use in the computer routine. With the characteristic B having been computed, then H is found by:

$$H = a + bB + cB^2 + dB^3 + eB^4 + fB^5 + gB^6 \quad (1)$$


where: a= 110.529; b= -1700.2762; c= 14226.658; d= -41673.495
e= 57423.444; f= -37223.825; g= 9235.1829

3 - Some Comments Thus Far

As may be inferred from the above, the steps for solving d.c. magnetic circuits are somewhat simple in principle yet will require a computer routine to do the tedious math required for convergence. If the pole geometry at the working air gap is other than a flat face the permeance of the gap becomes more complex to describe mathematically. Once written, a computer routine can document a full table of dimensions, power, flux, force vs. stroke, individual mmf drops (losses), total mass, plunger mass, inductance, and even first approximations of continuous power ratings and maximum power 'on' times at intermittent power levels. Additional BH equations can be developed or located for alternative materials such as cobalt and vanadium alloys. A high level computer language is not required; this author began solving problems such as this in 1982 on the early Commodore VIC-20 using 8 bits/byte and about 5k of memory.¹

4. - Modeling a Physical Model

In 1941 the publication *Electromagnetic Devices* by Herbert C. Roters was produced and began as a textbook for electrical engineering students at Stevens Institute of Technology. This book continued as a premiere reference source in solenoid magnetics for many years due to its detailed and exhaustive treatment of the subject at that time. The solenoid model below is taken from that source as an example for computing force (but with the benefit of the computer which Roters did not have). The computations will be treated with the simple routines of GWbasic or PCbasic for clarity.



¹ "Keep it as simple as possible but no simpler"; A. Einstein.

² Based on Roters' *Electromagnetic Devices*, vol.1, pp.73 & 232.

Fig. 2 Axial Solenoid Model²

The solenoid dimensions are shown both in inches and in MKS units (meters); only MKS units will be used in the computer routine of article 5.

aa	Case tube length	3.500 in.; 0.0889 m.
bb	Case tube o.d.	3.375 in.; 0.0857 m.
cc	Plunger and Stator Pole dia.	1.312 in.; 0.0333 m.
dd	Length of aft flux carrier	2.130 in.; 0.0541 m.
ee	I.d. of aft flux carrier	1.376 in.; 0.0350 m.
ff	Thickness of fwd. flux ring	0.328 in.; 8.33e-3 m.
gg	Length of Stator Pole	2.078 in.; 0.0528 m.
ss	Plunger stroke (gap) range	0.500 in. max.; 0.0127 m. max.
hh	Length of Plunger from aft flux carrier	1.572-ss in.; 0.03993-ss m.
kk	Case tube i.d.	3.125 in.; 0.079 m.
nn	Number of coil turns	2550
	Coil Current Increments	0.4, 0.8, 1.5 amps

5. A Bare-Bones Basic Routine

```

5 'Roters.bas mohler Aug. 1982 and May-Jun 2019
10 PI=3.14159:UO=.0000004*PI:Q=.00045:Z=0:M=39.37
20 AC=.0008234:AB=.0008722:AP=AB:LP=.0508
30 LC=.0889:LB=.0508:DP=.0333:RT=.038:RL=.0008128
40 PR=UO*PI*DP*RT/RL
50 FX=2040:' coil amp turns, also 1020 and 3825
60 FOR SS=.505 TO .015 STEP -.005:'inches
70 S=SS/M:'meters
80 PG=UO*AP/S:'main air gap permeance
90 B=Q/AC
100 GOSUB 300
110 FC=H*LC
120 B=Q/AB
130 GOSUB 300
140 FB=H*LB
150 B=Q/AP
160 GOSUB 300
170 FP=H*(LP-S)
180 FR=Q/PR
190 FG=Q/PG:'main air gap mmf
200 FT=FC+FB+FP+FR+FG:'total mmf drops
210 IF FT/FX>1.01 THEN Q=Q*.993:GOTO 90
220 IF FT/FX<.99 THEN Q=Q*1.012:GOTO 90
230 IF Z=0 THEN PP=PG:FF=FG:Z=1:GOTO 270
240 F=((FF+FG)/2)^2*(PG-PP)/2/((.005/M)
250 PP=PG:FF=FG:'holds the previous pp & ff for line 240.
260 PRINT "ss=";INT(SS*1000)/1000;"in.;" F=";INT(F*.2248*100!)/100!;"lbf.;" Q=";Q;" webers"
262 N=N+1:IF N=15 THEN PRINT "Hit any Key":N=0:X$=INPUT$(1):CLS
270 NEXT SS
280 END
300 A=110.529:Y=-1700.2762:C=14226.658:D=-41673.495:E=57423.444
302 F=-37223.825:G=9235.1829
304 H=A+Y*B+C*B^2+D*B^3+E*B^4+F*B^5+G*B^6:RETURN

```

Code keys: Lines 10-50 set physical constants and dimensions taken from Roters' example but in MKS units. Final stroke and derived force units revert to imperial units in keeping with the text.

Code symbols A_ are cross sectional flux areas; L_ are lengths; D_ are diameters; R_ are radial distances; P_ are permeances; FX is the coil amp-turns; F_ are amp-turns as computed; Q is flux (webers); SS is the solenoid stroke (air gap width). These symbols are not necessarily matched to the Fig.2 geometry.

The table of Fig. 3 below shows a comparison of calculated forces using the program above and those published by Roters for this model.³ Forces are in pounds.

Air Gap (in.)	Forces @ Coil mmf=1020		Forces @ Coil mmf=2040		Forces @ Coil mmf=3825	
	Text Data	Calculated	Text Data	Calculated	Text Data	Calculated
.500		0.75	3.0	3.0	14.1	10.6
.438		1.00	3.5	3.8	18.7	13.6
.375		1.32	4.5	5.2	22.8	18.5
.312		1.86	6.2	7.4	29.7	26.7
.250		2.8	12.4	11.4	41.1	40.2
.188	3.2	4.9	22.0	19.7	64.5	68.8
.125	10.4	10.3	44.7	41.3	112	139
.094	19.2	17.4	69.6	69.0	147	174
.063	40	35	109	129	200	212
.032	80	97	187	187	254	234
.016	135	155	251	205	289	244

Fig. 3 Text Book Data (from graph) and Calculated Data

The computer routine at line 10 sets an arbitrary starting flux ($\phi = Q$); then calculates the flux densities (B) of the iron segments, finds H for each in the subroutine, calculates their mmf drops ($\mathfrak{T} = H\ell$), calculates the air gap permeances ($P = \mu_0 A / \ell$) and their mmf drops ($\mathfrak{T} = \phi / P$). These are summed and compared to the coil mmf (FX). Lines 210 and 220 begin to adjust for the true flux value until the circuit and coil mmf values converge within 1%. The solenoid force is then found by $F = (\mathfrak{T}_g^2 / 2) / (dP / ds)$. The program steps in stroke increments of 0.005 in.; these increments are ds. If large stroke steps are made in the program then provision must be made to calculate the working air gap dP/ds at a shorter ds for each step.

The subroutine BH equation is monotonic from B>0.09 to B=2.4 and is for high iron content soft steels such as 1008, 1010, 1215, and 12L14. Other alloys, especially those containing permendur or vanadium will require a different curve. Likewise, sheet metal that has been hardened by rolling has a lower permeability and higher mmf losses.

³ Electromagnetic Devices by Herbert C. Roters; vol.1; 1941. References used by permission of the publisher, John Wiley & Sons, Inc.; (his Fig.3 on p.232)

The described method is not being recommended as a substitute for FEA when such software is available but gives some insight into how magnetics problems are solved. Prior to the computer, "solving" was approximate due to the tedious referral to a BH graph or lookup table to obtain H and then making the equation calculations (possibly many times depending upon the skill of the analyst to estimate the next trial flux level). This routine also does not address flux fringing but can be included with some study of fringing fields and the appropriate math (especially around the primary working air gap).

In this article the terms "ampere-turns drops", "mmf drops" and "losses" are synonymous and are used to itemize the series of magnetic potentials appearing around the circuit across each macro element for each static problem step; i.e., coil mmf input and plunger position.

It will be noticed from the table of Fig. 3 that the tested (Roters) and calculated (mine) forces cross over. The reasons, of course, cannot be determined at this late date. Possibilities are: steel quality, the force testing method, dimensional assumptions in the computer program, differing values of H vs. B. This article was not intended to prove accuracy but to show the application of the various magnetic identities leading to convergence. Similar computer routines with greater program detail can provide force, mass, and magnetic data to predict solenoid system response times to a high level of accuracy. The addition of spread sheet data collection and/or graphing would help to envision the distribution of mmf losses, force profile, etc.

In new development efforts a similar routine can optimize the ratio of steel and copper for any given power level and overall size whereby the force is observed to reach a maximum by incrementally trading steel volume and coil volume.

As noted in the introduction, the concept of magnetics is both interesting and mysterious. Some online inquirers are looking for that elusive formula that will easily and simply solve for solenoid force without factoring the non-linear BH characteristics into the solution. Solenoid efficiency is another elusive quality that will always remain low because the plunger kinetic energy is repeatedly discarded at the end of each stroke and lost to the heat of impact. (This is where motor efficiency excels). Coil winding efficiency is important but is not usually responsible for a significant improvement in performance. Solenoid efficiency runs about 15% to 22% (about the same as that of a '49 Ford). Solenoids utilizing high energy magnets in the plunger or armature will have higher efficiencies and other useful qualities.

Hopefully this discussion will aid in understanding a few of the principles for taming the solenoid. The mystery does not end here. Why and how do magnets attract? Why and how do they repel? Much has been said about electron spin, ether or no ether, magnetic domains, induction, and more. But when one experiences the tremendous pull or repulsion of rare earth magnets and electromagnets; those "forces at a distance"; the mystery remains.

6. – Related Thoughts

- a. Although flux is not considered as a "flow" thru a ferrous medium (Maxwell uses the term "propagation"), the mmf drops across circuit reluctances are analogous to voltage drops across electrical resistances. The summation of mmf drops and voltage drops are equal to the magnitude of their respective sources. Whereas electrical

current is a movement of charge in a conductor; i.e., electron flow, flux propagates by means of domain alignments in the iron molecules as dictated by the coil mmf in terms of polarity and magnitude. The magnetic circuit then is analyzed similarly to Ohm's law but labeled as Hopkinson's law. Solenoid and motor designers are concerned with focusing the flux to the working air gap so as to maximize the potential energy there where it converts to mechanical work. Ideally, the iron cross sectional area should not cause a restriction (reduced area) anywhere in the circuit unless purposefully at the working air gap for focusing purposes. This effort includes governing the flux density (B) in the main iron path to operate about or just above the "knee" of the BH curve for efficiency.

- b. At the working air gap of a solenoid, (where permeance $P = \mu_o A / L$) whether the gap is diminishing (L is the variable, Fig. 1) or the gap is constant (A is the variable, Fig. 4), the product of flux thru the gap and mmf drop across the gap determines the energy within the gap. This is the energy well from which mechanical work is derived. When L is the variable it can be seen that the change in energy will be exponential, as is common for most work solenoids, due to having L in the denominator of the permeance equation. When A is the variable, the change in energy is nearly linear consistent with A in the numerator. This makes possible proportional solenoids and solenoids that have a higher starting force (or torque) than their ending force. Here the slope of the force profile is a function of the rate of magnetic saturation in the steel (an engineered focusing or governing effect) as the displacement changes.




Fig. 4 Variable Area Air Gap; Constant Gap Length

- c. The BH tables for various magnetic core materials can usually be found but lack the corresponding equation from which to acquire the value of H for doing computer analysis. An equation to find H must be one in which as B increases, H also increases monotonically. If not, the computations may not converge and will loop endlessly. A couple of useful equation editors are CurveExpert and MicroMath's Scientist software. After deriving the equation, inspect its curve for accuracy and form. Minor equation modifications may be required if the curve near zero tends to go negative. If two or more equations are required to obtain a suitable fit with the data, inspect the results at the transition points of the equations for maintaining a slope consistent with the adjoining data.
- d. Inductance is a property that opposes a change of current in the inductor and is dependent upon the number of coil turns, instantaneous current, and circuit flux. It is

a calculated or measured result of the defined model and is not used as an input prior to solving the model. The inductance value is needful in calculating the current rise time and ultimate acceleration and response time of the solenoid when all other factors are factored in (moving mass, net force vs. position). Suffice it to say that the solenoid force is not instantaneous but ramps upward in concert with the current rise which then is negatively affected by the accelerated motion of the solenoid plunger, each of which are functions of the instantaneous inductance. This article does not address the calculations required to account for the negative going slope of the coil current resulting from the counter emf induced by the accelerating plunger. This factor, however, accounts for an otherwise erroneously rapid response time. Upon calculating a magnetic circuit, the necessary property values are usually also present from which to find the inductance from either of the equations:

$$L = N\phi / i \quad , \quad L = \mathfrak{L}\phi / i^2$$

- end -